

Exemplo 4:

Dados três valores, A, B, C, verificar se eles podem ser valores dos lados de um triângulo, e, se for, se é Δ qualquer, Δ equilátero ou Δ isósceles.

```

início
  real: A, B, C; (valores a serem verificados)
  leia (A, B, C);
  se A < B + C e B < A + C e C < A + B
  então
    se A = B e B = C
    então imprima ("TRIÂNGULO EQUILÁTERO");
    senão
      se A = B ou A = C ou C = B
      então imprima ("TRIÂNGULO ISÓSCELES");
      senão imprima ("TRIÂNGULO QUALQUER");
    fim se;
  fim se;
  senão imprima ("NÃO É TRIÂNGULO");
fim se;
fim
  
```

2.4 REGRAS PRÁTICAS PARA A CONSTRUÇÃO DE ALGORITMOS LEGÍVEIS

1. Procure incorporar comentários no algoritmo, pelo menos para descrever o significado das variáveis utilizadas. Comentários em PORTUGOL, podem ocorrer em qualquer parte do algoritmo, encerrados entre chaves {e}.

Ex.: {declaração de variáveis}, {cálculo da média}.

2. Escolha nomes de variáveis que sejam significativos, isto é, que traduzam o tipo de informação a ser armazenada na variável.

Ex.: NOTA
MATRÍCULA
CPF

3. Grife todas as palavras-chave (escritas com letras minúsculas) do algoritmo, destacando as estruturas de controle.

Ex.: A := 0;
se B \neq -1 então
:
:

4. Procure alinhar os comandos de acordo com o nível a que pertençam, isto é, destaque a estrutura na qual estão contidos.

Ex.: B := A * C;
enquanto P \neq 1 faça
P := P - 1;
F := X + NOTA/2;
leia (X, NOTA);
fim enquanto;
imprima (F, NOTA);
K := 1;
:

2.5 EXERCÍCIOS PROPOSTOS

```

1) início
   lógico: B1, B2, B3;
   se B1 então C1;
   senão
     se B2 então
       se B3 então C2;
       senão C3;
       C4;
     fim se;
   fim se;
   C5;
fim
  
```

Dado o algoritmo em PORTUGOL acima, responda:

- Se B1 = verdadeiro, B2 = verdadeiro, B3 = falso, quais os comandos que serão executados?
- Se B1 = falso, B2 = verdadeiro, B3 = falso, quais os comandos que serão executados?
- Quais os valores de B1, B2, B3 para que somente o comando C5 seja executado?
- Se B1 = falso, B2 = verdadeiro, B3 = verdadeiro, quais os comandos que serão executados?

```

2) lógico: A, B, C;
   real: X, Y;
   inteiro: V, L;
   A := falso; B := verdadeiro; C := falso;
   X := 1.5; Y := -3.2;
   X := X + 1;
   se C ou ((X + Y > 5) ou (não A e B)) então L := 0;
   senão L := 1;
   fim se;
  
```

Qual o valor de L após a execução deste trecho de algoritmo?

- Dado o algoritmo abaixo, que calcula o volume de uma esfera de raio R:

```

início
  real: VOLUME, PI, R;
  PI := 3,1416;
  R := 0;
  enquanto R < 6 faça
    VOLUME := 4/3 * PI * (R**3);
    imprima (R, VOLUME);
    R := R + 2;
  fim enquanto;
fim
  
```